
As you begin to get comfortable with the CA-Clipper language, the transition to network applications is a logical 
progression. In this article, I discuss the basics of multi-user programming using the various built-in features in 
Clipper. 
 
Before we begin, a short definition is in order. The term "network" used here involves this scenario: There 
exists a shared disk containing your application's data files (usually this is a hard disk residing on a dedicated 
file server) and a number of workstations running your application. The idea is to make the application aware of 
this environment and handle all of the potential problem areas that moving from a single-user to multi-user 
basis creates. If you use the built-in multi-user handling features Clipper provides, you don't have to worry 
about which LAN hardware or software environment you're running under. Your application will run under a file 
server-based system like NetWare or a peer-to-peer network like LANtastic. 
 
The problem domain 
 
One of the first questions that might occur to developers new to building multi-user apps is why networks 
programming is different from single-user programming. The answer to this question may not be immediately 
obvious until you consider what it means to be multi-user. 
 
First, consider a simple situation where a single-user program operates unaltered in a multi-user environment 
to see what dangers arise. Use 1 of an Employee Management System brings up an employee record edit 
screen and retrieves the record for "Jones." One of the fields on the screen is salary. At about the same time, 
user 2 also begins editing the "Jones" record. Each user is starting at the same screen. Now, user 1 cursors 
down to the salary figure and changes the amount to $32,000, saves the record and leaves the screen. User 2 
now unknowingly, changes the same salary field to $33,000 and also saves the record. In this case, since the 
record wasn't locked, the last value, $33,000, is the one inserted into the database and neither user knows 
what transpired. Multi-user control via record locking solves this concurrent update problem as well as many 
other potential problems. 
 
You might also ask what adding multi-user capabilities does to the complexity of the application and resulting 
code size. If you take the record locking route, the code may increase marginally (possibly 10 percent) due to 
the need to check for errors, display messages in the event of failed looks, etc. 
 
Remember, the primary goal of a multi-user application is to maintain the integrity of the data as many users 
vie for the same resources. 

 
File management 
 
A first step in giving your application that multi-user perspective involves opening the database. You have two 
choices, either exclusive (the default) or shared. Exclusive means that the operation you wish to perform on the 
file requires that no other user has access to it. Several commands need exclusive access. There are: 
 
REINDEX PACK ZAP 
 
In addition, certain reports should have exclusive access since, otherwise, the contents of the reports may be 
inconsistent. For example, an "Order Detail Report" which prints an Order Total for each customer might simply 
pick up total figures from an order master record. The detail items, however, might come from a transaction 
database. One possible problem might occurs if the report prints transactions 1, 2, and 3. Before the customer 
sub-total can be printed, another transaction is appended to the list by another user and the master record is 
updated. The report would show an overstated customer sub-total, based on four transactions instead of the 
three printed. 
 
To open file for exclusive access you would code: 
 
USE Order NEW EXCLUSIVE 
 

Although you can use the Xbase-style SET EXCLUSIVE ON command or a call to SET(_SET_EXCLUSIVE, 
1Toggle) , this practice is discouraged in favor of the EXCLUSIVE attribute of the USE command. To open a 

database for shared access you would code: 
 
USE (cTxnFile) NEW SHARED 
 
The NETERR() function normally follows attempt to open files on a network. The function returns a TRUE value 

in the event of a failed operation. Let's summarize the operations that lead to such a failure: 
 



If you issue a USE in SHARED mode and another user has exclusive access. 
 
If you issue a USE in EXCLUSIVE mode and another user has shared access. 
 

If you issue an APPEND BLANK and another user has the file locked with FLOCK() or has issued a simultaneous 
APPEND BLANK (since a successful APPEND BLANK issues a record lock automatically). 
 
After the program diagnoses these situations, it should elect to perform some "wait" philosophy, i.e., the 
requesting program should try again several times before giving up. One strategy is included in LOCKS.PRG 
described later. 
 
After opening a database for SHARED access with a FALSE NETERR() return value, you may proceed to open 
the appropriate orders: 
  
#define ORDERS_CUSTNO 'OrdCUST'  
#define ORDERS_PRODNO 'OrdProd'  
  
USE Orders NEW SHARED  
 
IF !NETERR()  
   * Add Orders to Order List  
   ORDLISTADD (ORDERS_CUSTNO)  
   ORDLISTADD (ORDERS_PRODNO)  
ELSE  
   ALERT ('Could not open Orders database')  
   RETURN  
ENDIF  

 
Most multi-user programs manipulate more than one file, so you need to develop a strategy for opening all of 
the files. For example, do you open them on an "as needed" basis? Or do you open them all at once? After 
doing some experimentation, I advise opening all the files at once and closing them all at exit. This simplifies 
the chore of making a file accessible when needed. 
 
There's also the consideration of when updates to a file are visible to other workstations, i.e., actually written to 
the disk. This question results from Clipper's use of buffers for database and index file updates. A current 
debate recommended that a "commit" operation follow the update and precede the unlock. The COMMIT 
command and two related database functions, DBCOMMIT() and DBCOMMITALL(), force the buffers to be 

written to disk. 
 
As a short example, we have the algorithm: 
 
IF DBSEEK (cWareHseNO) .AND. DBRLOCK (RECNO()) 
    Qty := nNewQty // Update record  
    DBCOMMIT() // FLush buffers  
    DBRUNLOCK (RECNO()) // Unlock record ENDIF  

 
Remember that shortcutting (assuming that /Z wasn't specified at compile time) avoids the DBRLOCK() call if 
DBSEEK() isn't successful. 

 
  



Locking mechanisms 
 
The nest area of concern relates to locking the application's various files and records to avoid convention 
difficulties. The bottom line is whenever you want to update a shared file, you must issue a lock first. You have 

two types of locks to choose from: a record lock or a file lock. Neither type prohibits another user from reading 
the file, but another user can not write to that file or record if locked. When you're finished with a lock, be sure 
to explicitly unlock the file or record even though Clipper goes through the unlocking motions when you close a 
database. Table 1 contains all the built-in Clipper locking mechanisms. 
  
Table 1--Network commands and their associated database  
 
functions (as mapped in STD.CH):  
 
Network commands:  
 

 
USE 
 
 

 
DBUSEAREA() 

 
 

 
Open a database SHARED or EXCLUSIVE.  

 
UNLOCK 
 
 

 
DBUNLOCK() 

 
 

 
Release any file or record locks in the current work area.  

 
UNLOCK ALL 
 
 

 
DBUNLOCKALL() 

 
 

 
Release all file in all work areas.  
 

      
Network functions:  
 

 
 NETERR() 

 

 
Check error status (as defined above).  

 

  
 RLOCK() 

 

 
Lock current record in the active work area.  

 

  
 FLOCK() 

 

 
Lock current file.  

 

  
 DBRLOCK() 

 

 
Lock a record based on identity.  

 

  
 DBRUNLOCK() 

 

 
Unlock a record based on identity or unlock all records.  

 

  
DBRLOCKLIST() 

 

 
Return array of identities for Lock List.  

 

 
New with version 5.2 is the ability to lock more than one record at once. To do this, the new database functions 
DRBRLOCK(), and DBRUNLOCK(), and DRBLOCKLIST() facilitate locking based on a record identities whereas 
the RLOCK() function simply locks the current record. To help maintain multiple locks, 5.2 includes the concept 
of a Lock List which you can think of as an array of identities tied to a database in a work area. In fact, the 
DRBRLOCKLIST() function returns just that, a single dimension array containing all locked identities. 
 
Both DBRLOCK() and DBRUNLOCK() have one parameter, the identity. The concept of an "identity" focuses 
around the need to uniquely identify a record in a database. In standard DBF files, the identity is the record 
number. With Clipper's expanded awareness of other data sources, due to its new compliment of RDDs, the 
identity concept had to be created. 
 
So for record locking you now have a choice. You may either use the new DBRLOCK() / DBRUNLOCK() or 
RLOCK() / DBRUNLOCK() / DBUNLOCKALL() combinations. The official word from Computer Associates is that 
all new code should use DBRLOCK() and DBRUNLOCK() since these functions handle both single locks (RLOCK() 
functionality and multiple locks. One caution if you decide to mix the two methods is that RLOCK() will first 
clear the Lock List before locking the current record. 
 
Now let's take a look at a simple example taking an existing single-user update code segment and making it 
network compliant. The underlying strategy is simple: lock the record, update it, and then unlock it. 
  



  * Single user code version  
 
USE Txn NEW  
SEEK cKeyValue  
IF FOUND()  
   Txn-> OrdeDate := DATE()  
ENDIF  
  
  * Multi-user coder version  
USE Txn NEW SHARED  
SEEK cKeyValue  
IF FOUND()  
   IF DBRLOCK (RECNO())  // Lock current record  
      OrderDate := DATE()  
      DRBUNLOCK (RECNO())  
   ELSE  
     ALERT ('Could not obtain lock')  
   ENDIF  
ENDIF  
 
As another example, the program below illustrates the Lock List capabilities of Clipper. The main program 
opens a database with the SHARED attribute (a requirement before locking occurs). Then checks NETERR() to 
see if the open succeeded. If so, it adds an order to the order list using ORDLISTADD(). 
 
It proceeds by calling the GroupLock() UDF and passing it the identities to lock. When using the default DBFNTX 
RDD, indentities are simply record numbers. Inside GroupLock(), step through the series of record numbers and 
use DBRLOCK() to add them to the Lock List. Return the Lock List array back to the main. Finally, show how 
individual identities in the Lock List may be removed with the DBRUNLOCK() function. 
 
Note that many database and Order Management system functions aren't defined on a work area without an 
open database. In these cases, the runtime error "Work area not in use" occurs. Some functions like EOF(), 
however, return defined values when performed on an empty work area. 
  
FUNCTION Main  
   LOCAL a LockList  
  
   USE Employee SHARED NEW  
 
   IF !NETERR ()  
      ORDLISTADD ('EmpNo')  
   ELSE  
      ALERT('File in use')  
   RETURN NIL  
ENDIF  
  
* Lock a group of records  
aLockList := GroupLock(3,5)  
  
DBRUNLOCK(4) // Remove one from lock list  
  
aLockList := DBRLOCKLIST() // Now  
 
// LEN(aLockList) = 2  
  
DBRUNLOCK() // Remove all remaining  
// from lock list  
 
DBCLOSEAREA()  
  
* 'Work area not in use' runtime error  
* aLockList := DBRLOCKLIST()  
  
RETURN NIL  
  
  



/***  
  
* GroupLock ( ,  )  
* -- right arrow] aLockList  

  
*/  
 
FUNCTION GroupLock (nFirstRec, nLastRec)  
  
   LOCAL nRecPtr  
  
   FOR nRecPtr := nFirstRec TO nLastRec  
      //Add identity to lock list  
      IF !DBRLOCK(nRecPtr)  
         DBRUNLOCK() // Unlock all  
      ENDIF  
   NEXT  
  
  RETURN DBRLOCKLIST()  // Return the lock list  

 
Record locking strategies 
 
There are two primary record locking strategies: pre-lock and post-lock. You must choose which works best for 
your application. The first approach says you should lock the record when users begin editing. The lock remains 
active until users complete the editing process. The strategy has the potential of keeping a record locked for a 
long time, especially if the user goes off to lunch while the record is on screen. The solution to the problem 
involves instituting a timer that terminates the editing session and automatically unlocks the record. Here's a 
sample prelock implementation: 
  
IF DBSEEK (cKeyValue) .AND. DBRLOCK(RECNO ())  
  
     Scatter() // Copy field values to array  
     EditRec() // Would include timer  
     * Confirm save here if you wish  
     Gather() // Copy array back out to fields  
     DBCOMMIT() // For safety  
     DBRUNLOCK(RECNO()) // Release from Lock List  
 
ENDIF  

 
The alternate approach is to first scatter the contents of the record's fields into an array, let the user edit the 
array and then lock the record before gathering back to the record. This solves the indefinite lock problem, but 
creates the situation where the record could have changed (by another user) between the scatter and gather 
operations. The solution to this problem involves comparing a saved copy of the original record to the version 
currently on disk. If they're different, you know someone else changed it while you were editing. Another way is 
to add a flag field to the database structure which indicates whether another user recently changed the record. 
Here's a sample postlock implementation: 
  
DBSEEK(cKeyValue) // Position to record  
Scatter() // Copy field values to array  
EditRec() // Edit array contents  
* Confirm save here is you wish  
* Determine if its safe to update here  
IF DBRLOCK (RECNO()) // Obtain lock  
   Gather() // Copy array back out to fields  
   DBCOMMIT() //For safety  
   DBRUNLOCK (RECNO()) // Release from Lock List  
ENDIF  

  
  



LOCKS.PRG  
 
To beef up the capabilities of Clipper's network programming features, the product ships with a PRG file found 
in the/CLIPPER5/SOURCE/SAMPLE directory that contains several useful UDFs that act as an interface for the 

built-in network functions. You should consider using these functions (or some of the publicly available 
derivatives of them) whenever multi-user enhancements to your code become necessary. 
 
One nice feature of all the UDFs in LOCKS.PRG is the timing of retries. On a busy network, it's possible that a 
network operation fails during the first few tries due to the activity of other users. Luckily, it's rather simple to 
provide for retries as evidenced in LOCKS.PRG. 
 
The optional parameter for the functions determines how many seconds you want the function to try the 
required operation. If you fail to specify a value, the default is two seconds as defined in the NET_SECS 
manifest constant found in LOCKS.PRG. If, however, you specify the value 0, the function keeps trying forever. 
The elapsed retry interval is defined by the constant NET_WAIT which has a value of 0.5. In these case, the 
functions return a .T. value if the operation was successful and .F. otherwise. 
  
 
AddRec() -> lSuccess  
  
     This function tries to append a blank record with an optional retry.  
  
FilLock()  -> lSuccess  

  
     FilLock() attempts to issue a successful file lock on the currently active database.  
  
NetUse(  <10penMode>  ) -> lSuccess  
  
    NetUse() tries to open the database in either EXCLUSIVE or SHARED mode.  The  parameter denotes the  
   file to open.  If  is .T. then the file is opened EXCLUSIVE and if .F. then SHARED.  
  
RecLock() -> lSuccess  
 
   This function tries to issue a rewcord lock for the current record. Note that RecLock() uses RLOCK() so 
multiple recored locks can't be activated. Here's an equivalent RecLock() function that handles multiple locks: 
  
#include "Common.ch"  
#define NET_WAIT 0.5 // Seconds to wait  between re tries  
#define NET_SECS 2 // Number of seconds to continue  retry  
 
FUNCTION Main  
   LOCAL a LockList    // Save the Lock List here.  
  
   USE Employee NEW SHARED  
  
  ? RecLock(4) // Add idewntity 4 to Lock List.  
  ? RecLock(6, 10) // Add identity 6 to Locko List.   
  aLockList := DBRLOCKLIST() // Result: {4,6}  
  
   DBRUNLOCK() // Remove all identities from Lock L ist.  
   DBCLOSEAREA()  
 
RETURN NIL  
  
  



/***  
* RecLock() -> lSuccess  
* Attempt to DBRLOCK() with optional retry  
* Parameters:  
*   nIdentity : Identity (recno) to add to Lock Lis t.  
*   nWaitSeconds : Optional time in seconds to retr y, operation default is NET_SECS  
*  
* Returns:  
*   .T. - Successful lock  
*   .F. - Unsuccessful lock  
*/  
  
FUNCTION RecLock( nIdentity, nSeconds)  
  
      LOCAL lForever  
  
      DEFAULT nSeconds TO NET_SECS  
  
     IF DBRLOCK(nIdentity)  
         RETURN (TRUE)  
     ENDIF  
  
     lForever := (nSeconds == 0) // Retry forever?  
  
     WHILE (1Forever .OR. (nSeconds > 0))  
         IF DBRLOCK(nIdentity) // Add identity to L ock List.  
            RETURN (TRUE) // Success.  
         ENDIF  
          INKEY (NET_WAIT)  
         nSeconds -= NET_WAIT  
    ENDDO  
 
RETURN (FALSE) // Lock not added to Lock List.  
  

Deadlock  
 
Deadlock is a common problem--contention for resources, either at the file or record level. Consider the 
situation where two users are running in a multi-user environment. Each user requires a resource which the 
other has locked. This is called "deadlock." As an example, consider two workstations running the same 
application. Workstation 1 successfully locks file A and needs file B to continue processing. At nearly the same 
time, workstation 2 begins another process that locks file B and needs file A. Both workstations wait for a file 
the other has locked. There is nowhere to go and the system hangs (assuming there are no time out provisions 
in the locking routine). 
 
The solution is to consistently open the files in the same order. That way, resource conflicts are settled 

immediately, before a deadlock can occur. This technique is called deadlock prevention. 
 
Note that since you may now lock more than one record in a single database via the Lock List, the potential for 
record-level deadlock exists, too. 
 
Testing multi-user apps 
 
As a beginning Clipper developer, you may be concerned about programming multi-user applications if you 
don't have a network on which to test them. This is an important concern since testing and debugging network 
software often takes considerable time since there are many combinations and sequences of events that may 
"break" the program. Luckily, there's a simple alternative to going out and buying expensive network software 
and hardware. All you need to do is simply run two instances of the same application in two different DOS 
windows under Windows. Using the inherent task-switching capability in Windows allows you to test and check 
all the various events that may occur. Make sure that SHARE is loaded and that each of the DOS box PIF files is 
set to run in "background" mode. 
 
That's all you need. Execute the first copy of your application in DOS box 1 (simulating user 1), then execute 
the second copy in DOS box 2 (user 2). You can switch between boxes, trying different features to use how 
they affect the other user. 

